

Kleincomputer KC85

Beschreibung zu M010 ADU

**veb mikroelektronik
>wilhelm pieck<
mühlhausen
im veb kombinat mikroelektronik**

1. Einleitung

Das Bindeglied zwischen den heutzutage meist digital realisierten Signalverarbeitungseinrichtungen und den analogen Signalen, d.h. wert- und zeitkontinuierlichen Signalen, sind Analog-Digital-Umsetzer. Mit der Verfügbarkeit hochintegrierter Digitalschaltkreise steigt die Nachfrage nach den entsprechenden Umsetzerbaugruppen. Durch die Entwicklung des Kleincomputers KC85/2 und seiner Nachfolgetypen rücken die Schnittstellen zur "analogen Umwelt" des Computers weiter in das Blickfeld der Anwender dieser Technik.

Der Modul M010 ADU1 (Analog-Digital-Umsetzer) realisiert eine vielfältig anwendbare Umsetzereinheit für den KC85/2 und seiner Nachfolgetypen.

Mit Hilfe des Moduls können vier Meßstellen abgefragt werden. Damit kann man Probleme der Meß-, Steuer- und Regelungstechnik sowie der Automatisierung im Labormaßstab realisieren.

In der Anleitung erfahren Sie, wie der Modul anzuschließen und zu bedienen ist. Weiterhin erhalten Sie wichtige Hinweise zur Arbeit mit dem Modul.

ACHTUNG!

Die an die Analogeingänge K1 bis K4 angeschlossenen Spannungen müssen die Anforderungen an die Sicherheitskleinspannung gemäß TGL 200-0602 erfüllen. Das heißt, daß die angeschlossene Wechselspannung oder Gleichspannung 25V bei einer Welligkeit kleiner 10% nicht überschreiten darf. Es ist durch geeignete Isolations- und/oder schaltungstechnische Maßnahmen dafür zu sorgen, daß dieser Wert nicht überschritten wird. Überschreitet die Spannung des speisenden Netzes diese Werte oder besteht die Möglichkeit zu deren Überschreitung, dann muß die Spannungsübertragung (vom speisenden Netz zur Sicherheitskleinspannung) über Transformatoren erfolgen, bei denen die Ausgangswicklung galvanisch von der Eingangswicklung getrennt ist. Das Isoliervermögen zwischen den genannten Wicklungen muß mindestens der Isolationsgruppe 5 nach TGL 20445/03 entsprechen. Das ist z. B. bei Schutztransformatoren nach TGL 200-1766/01 oder Spielzeugtransformatoren nach TGL 200-1773/01 der Fall.

2. Modulhandhabung

2.1. Stecken und Entfernen des Moduls

Der Analog-Digital-Umsetzer-Modul kann prinzipiell in jedem Modulsteckplatz betrieben werden, jedoch ist dabei die Modulpriorität in der gewählten Systemkonfiguration zu berücksichtigen. Es sind deshalb die nachfolgenden Informationen und angeführten Hinweise zu beachten.

Die Modulprioritätskette muß geschlossen bleiben. Also sind zuerst im Grundgerät der Steckplatz 8 (rechts), danach der Steckplatz C (links) und anschließend weitere Steckplätze von Erweiterungsaufsätzen in aufsteigender Reihenfolge zu belegen.

Achtung!

Das Stecken des Moduls bzw. das Entfernen des Moduls aus dem Steckplatz darf nur im ausgeschalteten Zustand des Systems erfolgen!

Damit ergeben sich folgende Handgriffe für das Stecken des Moduls:

- a) Den Computer ausschalten.
- b) Die Kappe des Modulschachtes abnehmen. Hierzu muß die Kappe an den gegenüberliegenden Griffflächen leicht zusammengedrückt werden.
- c) Den Modul bis zum fühlbaren Einrasten einschieben. Der Modul ist dann richtig kontaktiert, wenn sein hervorstehender Rand unmittelbar an der Gerätewand anliegt.
- d) Nun kann der Computer eingeschaltet werden.

Bevor die weitere Inbetriebnahme beschrieben wird, soll an dieser Stelle gleich das Entfernen des Moduls aus dem System erläutert werden. Zum Entfernen des Moduls sind folgende Schritte notwendig:

- a) Den Computer ausschalten.
- b) Den linken und rechten Zeigefinger unter den Modulkopf legen und mit den Daumen die seitlich am Modul befindlichen Hebel gleichzeitig nach unten drücken. Dabei rastet der Modul aus und wird etwa einen Zentimeter aus dem Gerät herausgeschoben. Nun kann der Modul leicht aus dem Schacht gezogen werden.
- c) Die Kappe des Modulschachtes auf die Schachtöffnung stecken.

2.2. Modulzuweisung und Steuerbyte

Der Digital-Analog-Umsetzer-Modul M010 hat im KC- System zwei Betriebszustände:

- INAKTIV Die Diode leuchtet nicht. Der Modul ist vom Prozessor getrennt.
- AKTIV Die Diode leuchtet. Falls es die Modulpriorität erlaubt, können über den aktivierten Modul Daten empfangen werden.

Der gewünschte Betriebszustand wird über das Kommando

SWITCH mm kk (Parameter durch Leerzeichen voneinander getrennt)

eingestellt. Die beiden Parameter dieses Kommandos realisieren:

- mm Mitteilung an das System, in welchem Modulschacht der zuzuweisende Modul kontaktiert ist.
Dabei ist die erste Stelle von mm die Nummer des Aufsatzes (im Grundgerät ist diese Stelle Null und kann weggelassen werden).
Die zweite Stelle von mm ist die Steckplatzadresse. Im Grundgerät existieren nur die Steckplatzadressen 8 (rechter Schacht) und C (linker Schacht).
- kk Schalten des Moduls. Die erste Stelle von kk ist Null und kann weggelassen werden.

Die zweite Stelle von kk kann sein:

- 0 - Modul inaktiv
- 1 - Modul aktiv.

Beispiel: Der Analog-Digital-Umsetzer-Modul M010 steckt im Schacht 8 des Grundgerätes und soll aktiviert werden. Es ist einzugeben:

SWITCH 8 1 .

Nach Betätigen der ENTER-Taste muß die Diode des M010 Analog-Digital-Umsetzer leuchten. Der Modul ist in den aktiven Zustand geschaltet. Auf dem Bildschirm erscheinen folgende Informationen:

08 E7 01 .

Mit dem ersten Byte wird angezeigt, daß der Modul im Steckplatz 8 angesprochen wurde. Das zweite Byte ist das Strukturbyte des Moduls, der im angesprochenen Modulschacht steckt. Durch das dritte Byte bestätigt der Computer die Eingabe des Parameters kk.

Jeder Modul enthält ein für ihn charakteristisches Strukturbyte. Dieses Strukturbyte widerspiegelt den Modultyp bzw. die innere Strukturierung des Moduls. Das Strukturbyte kann durch den Prozessor (auch im inaktiven Zustand des Moduls) gelesen werden. Dadurch kann sich der Nutzer jederzeit in einem ausgebautem System einen Überblick über die verfügbaren Module verschaffen und in Abhängigkeit davon seine Entscheidungen treffen.

Der Analog-Digital-Umsetzer-Modul besitzt, wie das Beispiel zeigt, das Strukturbyte:

E7H

Das Herausschalten des Moduls aus dem System (Betriebszustand INAKTIV) erfolgt über:

SWITCH mm 00 (für mm ist die Modulsteckplatzadresse einzugeben).

Auf dem Bildschirm erscheint:

mm E7 00

Hinweis: Nach Betätigen der RESET-Taste bleibt das Steuerbyte des Moduls erhalten. Ein aktiv geschalteter ADU-Modul ist weiterhin im aktiven Zustand. Lediglich der Schaltkreis PIO wird durch das Signal /RESET in den Anfangszustand versetzt und muß programmtechnisch neu initialisiert werden.

2.3. Modulprioritätskette

Jeder Analog-Digital-Umsetzer-Modul ist mit einer Prioritätssteuerung ausgestattet, die seine Eingliederung in das KC-System ermöglicht. Dadurch können mehrere ADU-Moduln in das KC-System eingeordnet werden.

Ist dies der Fall, gilt folgende Reihenfolge:

Derjenige aktivierte Modul, der sich auf dem Modulsteckplatz mit der niedrigsten Steckplatznummer befindet, besitzt gegenüber den anderen aktiv geschalteten Analog-Digital-Umsetzer-Moduln die höchste Priorität. Ein im Steckplatz 8 des Grundgerätes kontaktierter ADU-Modul ist danach höher priorisiert als ein ADU-Modul im Steckplatz C. Sind in beiden Schächten ADU-Moduln aktiviert, so kann der Modul im Schacht C nur dann benutzt werden, wenn der Modul im Schacht 8 INAKTIV geschaltet wird.

Verallgemeinert gilt:

Durch INAKTIV-Schalten des in der Rangfolge höherpriorisierten ADU-Moduls, kann der AKTIV geschaltete ADU-Modul mit der nächst niedrigeren Priorität genutzt werden.

Das AKTIV- bzw. INAKTIV-Schalten erfolgt über das SWITCH-Kommando (siehe Abschnitt 2.2.) bzw. über das entsprechende Betriebssystemunterprogramm.

3. Technische Beschreibung

3.1. Aufbau des ADU-Moduls

Der ADU-Modul besteht aus zwei funktionellen Baugruppen, der analogen und der digitalen Baugruppe. Die analoge Baugruppe besteht aus:

- Stromversorgung
- Analogmultiplexer
- Verstärkerbaugruppe mit Halteschaltung
- Analog-Digital-Umsetzer.

Die digitale Baugruppe gliedert sich auf in:

- Modulsteuerung
- Modulprioritätslogik
- Adreßdeko­der
- Ein/Ausgabe-Register mit ADU-Steuerlogik.

Der Modul M010 ADU1 ist so aufgebaut, daß er die multiplexe Erfassung von vier analogen Eingangssignalen ermöglicht. Die Frontplatte des M010 ist im Bild 1 dargestellt. Die hier verwendeten Bezeichnungen haben folgende Bedeutung:

- | | |
|-----------|--------------------------------------|
| K1 ... K4 | - Eingangskanäle |
| P1 und P2 | - Potentiometer |
| L | - Leuchtdiode (siehe Abschnitt 2.2.) |



Bild 1: Darstellung der Frontplatte des M010 ADU1

Mit Hilfe des Potentiometers P1 ist eine Eingangssignalverstärkung gemeinsam für alle Analogeingänge möglich. Damit kann die Auflösung der Eingangssignale vergrößert werden, wenn der maximal auftretende Endwert an den Meßkanälen

$$- 5V < UE < + 5V$$

beträgt.

Das Potentiometer P2 dient zur Nullpunkteinstellung.

Die beiden Potentiometer P1 und P2 lassen sich mit einem 2 mm Uhrmacherschraubendreher gut betätigen (siehe auch Abschnitt 5.).

3.2. Kurzbeschreibung der Schaltkreise PIO, ADU und Multiplexer

Den Hauptbestandteil des ADU-Moduls bilden:

- der Parallel Ein/Ausgabe-Schaltkreis PIO
(Parallel Input/Output Controller U855D)
- der Analog-Digital-Wandler ADU
(C570D für 8 bit Auflösung)
- 8-Kanal-Analog-Multiplexer/Demultiplexer (V4051D)

Der PIO ist ein hochintegrierter Schaltkreis des U880-Prozessor systems, der vorwiegend für den Einsatz in der Steuerungs- und Regelungstechnik vorgesehen ist. Der Schaltkreis ist weitgehend frei programmierbar und kann den peripheren Bedingungen angepaßt werden. Der PIO ermöglicht die parallele Ein- und Ausgabe von Daten zwischen CPU und der angeschlossenen Peripherie. Er enthält zwei TTL-kompatible 8-Bit-Tore (Ports), deren Betriebsart durch entsprechende Steuerbefehle programmtechnisch festgelegt werden kann. Der Schaltkreis verfügt über eine vollständige Logik zur Interruptkaskadierung und -priorisierung.

Weitere wichtige Merkmale des Schaltkreises sind:

- Interruptmöglichkeit im Quittungsbetrieb (handshaking) für schnelle Anforderungsbearbeitung. Die beiden 8-Bit-bidirektionalen Ports sind mit Einrichtungen für Quittungsbetrieb versehen.
- Programmierbare Betriebsarten
 - . Byte-Ausgabe (Mode 0)
 - . Byte-Eingabe (Mode 1)
 - . Byte-Ein/Ausgabe (bidirektionaler Betrieb, nur für Port A möglich (Mode 2))
 - . Bit-Ein/Ausgabe (Mode 3)
- Automatische Interruptvektorerzeugung und Prioritätskodierung durch Kaskadierung
- Alle Ein- und Ausgänge sind TTL-kompatibel.

Die Port-Signalbelegung des PIO-Schaltkreises zeigt die Tafel 1. Dabei ist Port A in Byteeingabe zu programmieren. Das Port B ist im Bitbetrieb zu betreiben, wobei die Bits 2 bis 6 auf Ausgabe zu programmieren sind (Maske 83H, siehe 5. Eichprogramm UP PIOIN).

Tafel 1: Signalbelegung des PIO-Schaltkreises

Port A:	A0	
	A1	
	A2	8-bit-Information
	A3	des ADU-Schaltkreises
	A4	(Umsetzwert)
	A5	
	A6	
	A7	
	ARDY	- mit B2=1 Start der Umsetzung
	ASTB	- aus /STS abgeleiteter Impuls (bei Interruptbetrieb des Moduls)
Port B:	B0	-
	B1	-
	B2	- Start der Umsetzung (mit ARDY=1)
	B3	-
	B4	- /A0 Kanaladresse
	B5	- /A1 für
	B6	- /A2 Multiplexer
	B7	- Statusausgang ADU (/STS)

Der Analog-Digital-Umsetzer ist ein monolithisch integrierter Schaltkreis mit einer signifikanten Wortlänge von 8 Bit. Der Wandler arbeitet nach dem Verfahren der sukzessiven Approximation. Durch den Schaltkreis können Eingangsspannungen im Bereich

$$-5V \leq UE \leq +5V$$

bzw.

$$0V \leq UE \leq 10V$$

verarbeitet werden.

In der im Modul angewandten Variante wird im bipolaren Mode gewandelt, d. h. bei einer Eingangsspannung von -5V ergibt sich der Wert 000H als Wandlungsergebnis, bei +5V der Wert 0FFH. Zur Steuerung der Analog-Digital-Umsetzung dient der Eingang L/S des ADU-Schaltkreises. Der Ausgang /STS signalisiert das Ende der Umsetzung. Nach erfolgter Umsetzung schalten die als Tri-State-Ausgänge ausgebildeten DATA-Ausgänge vom hochohmigen in den aktiven Zustand. Die AD-Umsetzung wird mit dem Übergang des Signals L/S von "H" nach "L" gestartet. Nach der Umsetzzeit schaltet der Ausgang /STS von "H" nach "L". Die DATA-Ausgänge tragen jetzt die dem analogen Eingangssignal entsprechende digitale Ausgangsinformation.

Nachfolgend sollen die Möglichkeiten der Umsetzung beschrieben werden.

- Start der Umsetzung:

Die Umsetzung des ADU kann auf unterschiedliche Art gestartet werden. Dazu sind die Leitungen ARDY und B2 des PIO-Schaltkreises NAND verknüpft.

Eine Startmöglichkeit besteht darin, durch Setzen des Bits 2 von Port B den Umsetzvorgang einzuleiten. Dazu muß jedoch der RDY-Ausgang des Port A "H"-Pegel führen. Dies kann durch eine Pseudoeingabe des auf Byteeingabe programmierten Port A bei der Initialisierung des PIO-Schaltkreises erreicht werden (siehe /1/). Nach dem Setzen von Bit 2 sollte unmittelbar eine Ausgabeoperation folgen, die dieses Bit wieder rücksetzt. Damit kann zielgerichtet im Pollingbetrieb eine Meßstellenabfrage erfolgen. Der Umsetzvorgang des ADU-Schaltkreises kann dabei durch Bit 7 von Port B (/STS) kontrolliert werden.

Um eine höhere Umsetzrate zu erreichen, bietet sich ARDY als Auslösesignal an. Dabei wird Bit 2 von Port B auf "H" programmiert. Nach Ablauf der Umsetzung schaltet ARDY auf "L". Das ADU-Datenbyte ist im Port A eingeschrieben und kann durch die CPU abgeholt werden. Nach einem Eingabebefehl geht das Signal ARDY auf "H". Damit wird der nächste Umsetzvorgang ausgelöst.

- Datenübernahme:

Hier bestehen ebenfalls zwei Möglichkeiten. Der Statusausgang des ADU-Schaltkreises (/STS) ist einmal mit dem Bit B7 und zum anderen über einen Monoflop mit dem ASTB-Eingang verbunden. Somit kann man die Übernahme einmal über das Bit B7 steuern (Pollingbetrieb) und andererseits durch die Auslösung eines Interrupts über den ASTB-Eingang. Mit dieser Übernahme werden die höchsten Umsetzraten erreicht.

- Ablauf der Umsetzung:

Zunächst muß über die Bits B4 bis B6 des PIO-Schaltkreises ein Kanal über den Multiplexer ausgewählt werden. Nach einer Aufladezeit des Hold-Kondensators von 600 ns kann der Start der Umsetzung in der beschriebenen Art erfolgen.

Damit wird der Hold-Kondensator vom ausgewählten Kanal getrennt und die Umsetzung beginnt. Durch die Abtrennung des Kondensators kann sofort nach




Bild 2: Taktdiagramm der Meßwertumsetzung

dem Start der Umsetzung ein neuer Kanal des Multiplexers ausgewählt werden. Das ist besonders bei hohen Umsetzraten von mehreren Eingangskanälen zu empfehlen. Nach der Datenübernahme kann dann sofort die Umsetzung mit dem neu eingestellten Kanal beginnen.

Der Multiplexerschaltkreis enthält einen achtkanaligen Multiplexer/Demultiplexer. Von den acht Kanälen Z0 bis Z7 wird über digitale Steuereingänge A0, A1, A2 ein Kanal ausgewählt und auf den Ausgang Y geschaltet.

Im Modul sind vier Analogeingänge über den integrierten Analogmultiplexer einzeln auswählbar. Der maximale Eingangsspannungsbereich beträgt +5V ... -5V.

Die Eingangssignale des Analogmultiplexers zeigt Tafel 2.

Tafel 2: Eingangssignale des Analogmultiplexers

PIO			Kanal- adresse	Steuerbyte		Signal
B6	B5	B4		hex.	dez.	
1	1	1	0	70	112	Kontrolle VN
1	1	0	1	60	96	Kontrolle VP
1	0	1	2	50	80	Referenzspannung
1	0	0	3	40	64	Analogmasse
0	1	1	4	30	48	Eingangskanal 1
0	1	0	5	20	32	Eingangskanal 2
0	0	1	6	10	16	Eingangskanal 3
0	0	0	7	00	0	Eingangskanal 4

Die Spannungswerte der Kanaladressen 0 und 1 dienen lediglich zur qualitativen Kontrolle der Stromversorgung des Moduls.

Die Referenzspannung der Kanaladresse 2 kann als Bezugsspannung herangezogen werden. Damit ist es möglich, beispielsweise Temperaturdriften programmtechnisch zu erfassen und zu korrigieren.

3.3. Interruptprioritätskette

Der ADU-Modul kann voll interruptgesteuert im KC-System betrieben werden.

Es ergibt sich folgende Prioritätskette für die Interruptauslösung:

PIO,PORT A -- > PIO,PORT B.

Wird mit mehreren Moduln gleichen Typs gearbeitet, ist zu beachten, daß die niederpriorisierten Moduln einen Interrupt auslösen können (bei entsprechender Initialisierung) und dieser von der CPU erkannt und angenommen wird. Ein Datenaustausch ist aber nur mit dem höchstpriorisierten ADU-Modul möglich. Durch eine entsprechende Wahl des Interruptvektors ist zu sichern, daß die aktivierte Interruptroutine höherpriorisierte Moduln inaktiviert, um den Datenaustausch mit dem Interrupt anmeldenden Modul zu ermöglichen, da sonst kein Bezug zum beabsichtigten Eingangssignal möglich ist.

3.4. Interne Adressen

Für den Analog-Digital-Umsetzer-Modul M010 sind im KC-System die in Tafel 3 angegebenen Adressen reserviert.

Tafel 3: Interne Adressen des ADU-Moduls

Ein/Aus- gabekanal	PIO- Kanal	Bedeutung
40H	A	Datenwort
41H	B	Datenwort
42H	A	Steuerwort
43H	B	Steuerwort

3.5. Anschlußbedingungen

Für die Eingangssignale von allen vier Kanälen des ADU-Moduls gilt der Wertebereich von +5V bis -5V.

3.6. Eingangsschutz

Alle Anschlußsignale sind mit einer Schutzschaltung versehen. Diese Schaltung gewährleistet einen Überspannungsschutz der Schaltkreise in einem Spannungsbereich von

+25 V bis -25 V.

Eine Überlastung beeinflusst dabei alle Kanäle unabhängig vom Entstehungsort und sollte deshalb vermieden werden.

4. Zusammenfassung

Der Analogeingabemodul ermöglicht die multiplexe Erfassung von vier Eingangskanälen. Durch eine regelbare Verstärkung und regelbaren Offset sind die Eingangssignale an den Arbeitsbereich des ADU1 optimal anzupassen. Auftretende Drifterscheinungen können durch zwei Referenzsignale erfaßt und korrigiert werden. Zum Betrieb des Moduls sind keine externen Spannungen erforderlich. Bei der Arbeit mit dem Modul hat es sich als zweckmäßig erwiesen, wenn der Modul ca. 10 min. lang im eingeschalteten System steckt. In dieser Zeit hat sich der Modul entsprechend erwärmt und eine gute Funktionsweise kann garantiert werden. Die wichtigsten technischen Daten enthält die Tafel 4.

Tafel 4: Technische Daten des Analogeingabemoduls M010 ADU1

A-D-Umsetzer	C570D (8 bit)
Linearitätsfehler	maximal + 1 LSB
Umsetzzeit	maximal 40 μ s, typisch 25 μ s
Anzahl der Eingangskanäle	4
Strukturbyte	0E7H
PIO-Adressen	040H Port A Datenwort 041H Port B Datenwort 042H Port A Steuerwort 043H Port B Steuerwort
Eingangsspannung UE in V	$\leq + 6V$, typ. + 5V
Eingangswiderstand RE in k Ω	typ. > 100
Überspannungsschutz in V	$\leq + 25V$
Verstärkung	regelbar, gemeinsam für die 4 Meßkanäle
Offset	regelbar, gemeinsam für die 4 Meßkanäle
minimale Eingangsspannung bei maximaler Auflösung	500 mV
Besonderheiten	- keine externe Spannungsversorgung - zwei Eichsignale zur Erfassung von Driftsignalen . Referenzsignal über B589Nm (1,2V ... 1,25V, TK=0 ... 10-6/K) . Analogmasse

5. Eichprogramm

Der Modul M010 ADU1 wird vom Hersteller so abgeglichen und geeicht ausgeliefert, daß ein Digit 40 mV entspricht (maximale Auflösung bei einem Eingangssignal von 5V). Mit Hilfe des Eichprogrammes ist dem Anwender die Möglichkeit gegeben, zunächst einmal die wichtigsten Spannungswerte zu überprüfen. Es werden auf dem Bildschirm folgende Spannungen hexadezimal angezeigt:

- qualitativer Kontrollwert VP (0D6H)
- qualitativer Kontrollwert VN (02AH)
- Referenzspannung 1,2V ... 1,25V (09FH ... 0A1H)
- Analogmasse (07FH)
- Eingangsspannung der Kanäle 1 bis 4 (K1 ... K4).

Mit Hilfe der Potentiometer P1 und P2 (siehe Bild 3) können die Referenzspannung und die Analogmasse auf die Standardwerte 0A0H und 080H eingestellt werden. Für die Arbeit mit dem Modul bedeuten diese Werte, daß bei einem Eingangssignal von + 5V die höchste Auflösung (0FFH) gegeben ist.




Bild 3: Darstellung der Potentiometerbetätigung

Liegen nun an den einzelnen Eingangskanälen des Moduls geringere Spannungswerte als + 5V an, so kann mittels Potentiometer P1 die Auflösung erhöht werden. Anschließend muß natürlich eine Normierung über Software mittels innerer oder äußerer Referenz vorgenommen werden. Hierzu soll das folgende Beispiel erläuternd dienen.

Es sei gegeben, daß folgende Spannungen an den Kanälen K1 bis K4 anliegen:

- K1 = 1,4V (0A4H)
- K2 = 0,2V (085H)
- K3 = 2,6V (0C3H)
- K4 = 1,2V (09FH).

Wie man leicht an den Spannungswerten erkennen kann, erfolgt die Umsetzung nicht mit maximal möglicher Auflösung. Man erhöht nun so lange die Verstärkung (Potentiometer P1), bis für den Kanal K3 ein Spannungswert von 0FFH angezeigt wird. Somit ist für die angenommene Eingangsspannungsbelegung die maximale Auflösung eingestellt.

Die Eingabe des aufgeführten Eichprogrammes kann wie folgt geschehen. Steht ein DEVELOPMENT-Modul M027 zur Verfügung, so kann das Listing eingegeben und übersetzt werden. Andernfalls kann auch die Eingabe des Maschinencodes mittels CAOS-Kommando 'MODIFY' ab Adresse 0200H erfolgen. Hierbei ist allerdings darauf zu achten, daß von allen Texten im Assemblerlisting nur die Codes der ersten vier Textzeilen aufgeführt sind. An der Adreßzählung ist erkennbar, daß die anderen Textzeilen auch im Speicher abgelegt werden, wenn der Assembler Quelltext vom Assembler selbst übersetzt wird.

Bei der Programmeingabe über 'MODIFY' dürfen die im Listing fehlenden Textcodes nicht vergessen werden, weil anderenfalls das Programm nicht lauffähig ist. Unabhängig von der Art der Eingabe wird das Programm mit dem Namen ADU1 aufgerufen. Zum Abspeichern auf Kassette ist folgende Kommandoeingabe notwendig:

```
SAVE 200 33D
NAME:ADU1
```

Durch Betätigen der <BRK>-Taste verläßt man das Eichprogramm und kehrt in das CAOS-Menü zurück.

Es ist darauf zu achten, daß der Modul mit der Anweisung

```
SWITCH mm kk
```

(siehe Abschnitt 2.2.) in den aktiven Zustand versetzt wird, bevor das Programm mit ADU1 aufgerufen werden kann.

Alle weiteren Informationen zum Programm können dem Listing entnommen werden.

Eichprogramm:

```
3000
3000
3000
3000      ;   E I C H P R O G R A M M   A D U 1
3000
3000      ;           12.06.87   A U M Y   M U E H L H A U S E N
3000
3000      ;VEREINBARUNGEN:
3000      COLOR   EQU    0B7A3H   ;FARBBYTE
3000      VERT    EQU    0F003H   ;UP-VERTEILER1
3000      PAD     EQU    40H      ;PIO KANAL-
3000      PBD     EQU    41H      ; ADRESSEN
3000      PAC     EQU    42H
3000      PBC     EQU    43H
3000
3000      ORG     200H
0200 3AA3B7   COLWE: LD      A, (COLOR)
0203 EE48    XOR     48H
0205 32A3B7   LD      (COLOR), A
0208 C9      RET
0209
0209 7F7F    DEFW   7F7FH
020B 41445531 DEFM   'ADU1'
020F 01      DEFB   1
0210 CD0002   ADU:  CALL   COLWE
0213 CD03F0   CALL   VERT
0216 23      DEFB   23H
0217 0C0A    DEFW   0A0CH
0219 0A0A    DEFW   0A0AH
021B 0A0A    DEFW   0A0AH
021D 20202020 DEFM   '   E I C H P R O G R '
0233 41204D20 DEFM   'A M M   A D U 1'
0242 0A0D    DEFW   0D0AH
0244 0A0A    DEFW   0A0AH
0246 0A0A    DEFW   0A0AH
0248 20565020 DEFM   ' VP   VN   REF   MASSE '
025C 204B3120 DEFM   ' K1   K2   K3   K4 '
026E 0D0A    DEFW   0A0DH
0270 0A00    DEFW   000AH
0272 CD0002   CALL   COLWE
0275 CDCB02   CALL   PIOIN   ;INITIALISIERUNG
0278 CD03F0   LO1:  CALL   VERT
027B 2B      DEFB   2BH      ;SPACE
```

027C	3E60		LD	A, 60H	;VERS.POS-KANAL
027E	CD0803		CALL	UMSET	;UMSETZUNG ADU
0281	CD0202		CALL	ANZEI	;WERT-AUSGABE
0284	3E70		LD	A, 70H	;VERS.NEG
0286	CD0803		CALL	UMSET	
0289	CD0202		CALL	ANZEI	
028C	3E50		LD	A, 50H	;REFERENZ
028E	CD0803		CALL	UMSET	
0291	CD0202		CALL	ANZEI	
0294	3E40		LD	A, 40H	;MASSE
0296	CD0803		CALL	UMSET	
0299	CD0202		CALL	ANZEI	
029C	3E30		LD	A, 30H	;K1
029E	CD0803		CALL	UMSET	
02A1	CD0202		CALL	ANZEI	
02A4	3E20		LD	A, 20H	;K2
02A6	CD0803		CALL	UMSET	
02A9	CD0202		CALL	ANZEI	
02AC	3E10		LD	A, 10H	;K3
02AE	CD0803		CALL	UMSET	
02B1	CD0202		CALL	ANZEI	
02B4	3E00		LD	A, 0	;K4
02B6	CD0803		CALL	UMSET	
02B9	CD0202		CALL	ANZEI	
02BC	CD03F0		CALL	VERT	
02BF	230B		DEFW	0B23H	
02C1	1900		DEFW	0019H	;CTBL
02C3	CD03F0		CALL	VERT	
02C6	2A		DEFB	2AH	;BRK?
02C7	D27802		JP	NC, LO1	
02CA	C9		RET		
02CB	F3	PIOIN:	DI		
02CC	3E4F		LD	A, 4FH	
02CE	D342		OUT	PAC	;BYTE-IN
02D0	3E03		LD	A, 3	
02D2	D342		OUT	PAC	;INT SPERREN
02D4	3ECF		LD	A, 0CFH	
02D6	D343		OUT	PBC	;BITBETRIEB
02D8	3E83		LD	A, 83H	
02DA	D343		OUT	PBC	;MASKE I/O
02DC	3E17		LD	A, 17H	
02DE	D343		OUT	PBC	;INT KONTROLLE
02E0	3E7F		LD	A, 7FH	
02E2	D343		OUT	PBC	;MASKE INT
02E4	3E10		LD	A, 10H	
02E6	D343		OUT	PBC	;INT-VEKTOR
02E8	DB40		IN	A, PAD	;PSEUDO-IN
02EA	212D03		LD	HL, SERV1	
02ED	221001		LD	(0110H), HL	;IV
02F0	FB		EI		
02F1	C9		RET		
02F2	3A3D03	ANZEI:	LD	A, (WERT)	
02F5	CD03F0		CALL	VERT	
02F8	1C		DEFB	1CH	;DRUCK VON A
02F9	CD03F0		CALL	VERT	
02FC	23		DEFB	23H	
02FD	202020		DEFM	' '	
0300	00		DEFB	0	
0301	3E0A		LD	A, 10	
0303	CD03F0		CALL	VERT	
0306	14		DEFB	14H	;ZEITSCHLEIFE
0307	C9		RET		
0308	F3	UMSET:	DI		
0309	D341		OUT	PBD	;KANAL EINSTELL.

```

030B F5          PUSH    AF      ;KANAL RETTEN
030C 3E83       LD      A,83H
030E D343       OUT     PBC      ;INT FREI
0310 3E01       LD      A,1
0312 323E03    LD      (ZAEHL),A;EINMAL UMSETZ.
0315 F1         POP     AF
0316 CBD7       SET     2,A      ;STARTBIT SET
0318 D341       OUT     PBD      ;START!
031A CB97       RES     2,A
031C D341       OUT     PBD      ;STARTBIT RES
031E FB         EI
031F 3A3E03    W1:    LD      A,(ZAEHL)
0322 B7         OR      A        ;INT ERFOLGT?
0323 20FA       JR     NZ,W1    ;NOCH NICHT
0325 3E03       LD      A,3
0327 D343       OUT     PBC      ;INT SPERREN
0329 3A3D03    LD      A,(WERT)
032C C9         RET
032D 08         SERV1: EX     AF,AF
032E D9         EXX
032F 213E03    LD      HL,ZAEHL
0332 35         DEC     M        ;INT IST ERFOLGT
0333 DB40       IN     A,PAD    ;8 BIT EINLESEN
0335 323D03    LD      (WERT),A;ABSPEICHERN
0338 D9         EXX
0339 08         EX     AF,AF
033A FB         EI
033B ED4D       RETI
033D          PUFFER: EQU    #
033D          WERT    EQU    PUFFER
033D          ZAEHL   EQU    PUFFER+1

```

6. Anwendungsbeispiel

a) Für den Anwender wird ein einfaches Beispiel in BASIC vorgegeben. Anhand dieses Demonstrationsbeispiels lassen sich gut die Funktionsweise und die Handhabung des M010 ADU1 Moduls erkennen.

Es wurde bewußt ein simples Beispiel mit geringem Aufwand an Schaltungs- und Programmtechnik angegeben. Für die kleine Schaltung werden zwei Bausteine aus dem SEG Elektrotechnik/Elektronik/Automatisierungstechnik vom VEB Kombinat Polytechnik und Präzisionsgeräte Karl-Marx-Stadt verwendet. Es sind dies die Bausteine Nr. 31 (Spannung stabilisator) und Nr. 38 (Potentiometer).

In das im Bild 4 angegebene Blockschaltbild kann an die Klemmen A und B ein Voltmeter zusätzlich zur analogen Anzeige angeschlossen werden.




Bild 4: Blockschaltbild für Anwendungsbeispiel

```

10 !          BEISPIELPROGRAMM  ADU1
20 !
30 !   25.09.87  AUMY  MUEHLHAUSEN
40 !
50 !INITIALISIERUNG:
60 SWITCH 8,1: ! ADU IM SCHACHT 8
70 WINDOW0,31,0,39:CLS
80 OUT66,127: !PIO A BYTE-EINGABE
90 OUT67,207: !PIO B BIT-BETRIEB
100 OUT67,131: !I/O MASKE
110 OUT65,52 : !KANAL1 AUSWAEHLEN UND STARTBIT SETZEN
120 !
130 !BILD ZEICHNEN:
140 X=20
150 LINE20,128,310,128,7
160 LINE20,10,20,245,7
170 PRINT AT(0,0); "U/V"
180 WINDOW3,30,0,1
190 PRINT" 4      3      2      1      0      -1      -2      -3      -
4"
200 PRINT AT(17,9); "2      4      6      8      t/s"
210 LINE19,231,21,231,7
220 LINE19,206,21,206,7
230 LINE19,180,21,180,7
240 LINE19,154,21,154,7
250 LINE19,102,21,102,7
260 LINE19,77,21,77,7
270 LINE19,51,21,51,7
280 LINE19,25,21,25,7
290 LINE76,127,76,129,7
300 LINE132,127,132,129,7
310 LINE188,127,188,129,7
320 LINE244,127,244,129,7
330 !
340 !TASTATURABFRAGE:
350 PRINTAT(30,38);COLOR21,1;CHR$(141)

```

```

360 E=PEEK(509):IFE<>13THEN360
370 PRINTAT(30,38);" "
380 !
390 !MESSEN UND ZEICHNEN:
400 FOR I=1 TO 280
410 A=INP(64):! MESSWERT HOLEN UND UMSETZUNG NEU STARTEN
420 X=X+1
430 PSETX,A
440 FORZ=1TO7:NEXT
450 NEXT
460 !
470 !TASTATURABFRAGE:
480 PRINTAT(30,38);COLOR21,1;CHR$(141)
490 E=PEEK(509):IFE<>13THEN490
500 GOTO70

```

Die Schaltung kann selbstverständlich auch mit anderen Mitteln ähnlich aufgebaut werden. Es ist aber stets darauf zu achten, daß der typische Eingangsspannungsbereich des ADU-Moduls von +5V nicht überschritten wird. Nachdem die Schaltung aufgebaut und das kurze BASIC-Programm implementiert ist, läßt sich gut die Arbeitsweise des Moduls erkennen. In dem hier vorgeschlagenen Beispiel wurde der Kanal 1 verwendet. Wenn auf dem Bildschirm das Diagramm vollständig gezeichnet ist, werden nach Betätigen der <ENTER>-Taste für ca. 10s Werte eingelesen und angezeigt. Durch drehen des Potentiometers (SEG Baustein Nr. 38) kann der Wertebereich von 0 ... 5V beliebig verändert werden.

Für einen Spannungswert von 0V gibt der ADU-Modul einen Wert von 07FH und für 5V einen Wert von 0FFH an den Computer aus, der hier direkt als Y-Koordinate verwendet wird.

- b) Das nachfolgend dargestellte Assemblerprogramm zeigt die Möglichkeit zur ständigen Abfrage einer Meßstelle im Polling-Betrieb. Die Meßwertumsetzung wird dabei durch das Signal RDY (RDYA = 1) gestartet. Nach jeder Einleseoperation des Ports A durch die CPU erfolgt damit der Start zur nächsten Umsetzung. Der Meßwert steht im Register A für die Weiterverarbeitung zur Verfügung. Im angegebenen Beispiel erfolgt die Ausgabe des Meßwertes auf den Bildschirm. Das Kanalsteuerwort KST definiert den Eingangskanal und setzt Bit 2 von Port B. Durch Variation des Steuerwortes im Programm können auch andere Meßkanäle aktiviert werden. Die Programmteile zur PIO-Initialisierung und Meßwernerfassung können als Richtlinie für eigene Programmentwicklungen dienen. Die Eingabe dieses Programmes erfolgt nach den Erläuterungen im Abschnitt 5.

Beispielprogramm:

```

3000 ;*****
3000 ;
3000 ;           BEISPIELPROGRAMM
3000 ; STAENDIGE ABFRAGE DES MESSKANALS K1
3000 ; IM POLLING-BETRIEB
3000 ; (START DER UMSETZUNG MIT RDYA=1)
3000 ;
3000 ;AUFRUF UEBER MENUFUNKTION:
3000 ;ADU STECKPLATZ (Z.BSP.:ADU 8)
3000 ;
3000 ;FUNKTIONEN:
3000 ;-PIO-INITIALISIERUNG
3000 ;-START DER MESSWERTUMSETZUNG
3000 ;-EINLESEN DES MESSWERTES
3000 ;-AUSGABE DES MESSWERTES AUF BILDSCHIRM
3000 ;
3000 ;*****
3000 ;
3000 VERT     EQU     0F003H
3000 PAD     EQU     40H      ;PORT A DATEN
3000 PBD     EQU     41H      ;PORT B DATEN
3000 PAC     EQU     42H      ;PORT A STEUER
3000 PBC     EQU     43H      ;PORT B STEUER
3000 KST     EQU     34H      ;K1,B2=1
3000 ;
3000 7F7F          DEFW     7F7FH
3002 414455      DEFM     'ADU' ;MENUFUNKTION
3005 01          DEFB     1
3006 3E02        LD      A,2
3008 1601        LD      D,1
300A 5A          LD      E,D
300B CD03F0      CALL    VERT   ;MODUL ON
300E 26          DEFB     26H
300F 3EE7        LD      A,0E7H ;STRUKTURBYTE
3011 BC          CP      H
3012 2805        JR      Z,AD1
3014 CD03F0      CALL    VERT
3017 19          DEFB     19H   ;ERROR
3018 C9          RET
3019 ;
3019 CD03F0      AD1     CALL    VERT
301C 23          DEFB     23H
301D 0C          DEFB     0CH
301E 0A0A        DEFW     0A0AH
3020 00          DEFB     0
3021 CD3830      CALL    INIT   ;INIT PIO
3024 3E34        LD      A,KST  ;STEUERWORT
3026 CD4A30      CALL    UMSET  ;STARTE ADU
3029 CD03F0      AD2     CALL    VERT   ;UP MESSWERTVER-
302C 1C          DEFB     1CH   ;ARBEITUNG (HIER
302D CD03F0      CALL    VERT   ;NUR AUSGABE AUF
3030 23          DEFB     23H   ;BILDSCHIRM)
3031 19          DEFB     19H
3032 00          DEFB     0
3033 ;
3033 CD4C30      CALL    UMSET1  ;LESE NEUEN
3036 ;           ;MESSWERT, STARTE
3036 ;           ;ADU
3037

```

```

3038 ;
3038 ;*****
3038 ;
3038 ;          PIO-INITIALISIERUNG
3038 ;
3038 ;*****
3038 ;
3038 3ECF  INIT    LD      A,0CFH  ;PORT B,BIT E/A
303A D343          OUT    PBC,A
303C 3E83          LD      A,83H  ;PORT B,I/O-MAS-
303E D343          OUT    PBC,A  ;KE
3040 3E4F          LD      A,4FH  ;PORT A,BYTE-
3042 D342          OUT    PAC,A  ;EINGABE
3044 AF            XOR     A          ;AUSGABEBITS=0
3045 D341          OUT    PBD,A
3047 DB40          IN     A,PAD  ;PSEUDOEINGABE
3049                      ;RDYA=1
3049 C9            RET
304A ;
304A ;*****
304A ;
304A ;          UMSETZUNG
304A ;
304A ;*****
304A ;
304A D341  UMSET  OUT    PBD,A  ;AUSGABE KST
304C                      ;STARTE UMSET-
304C                      ;ZUNG MIT B2=1
304C DB41  UMSET1 IN     A,PBD
304E CB7F          BIT    7,A   ;/STS=0
3050 20FA          JR     NZ,UMSET1
3052 DB40          IN     A,PAD  ;LESE MESSWERT,
3054                      ;AKKU=MESSWERT,
3054                      ;STARTE NAECHSTE
3054                      ;UMSETZUNG MIT
3054                      ;RDYA=1
3054 C9            RET
3055 ;

```

7. Zubehör

Als Zubehör zum ADU-Modul erhalten Sie vier HF-Stecker 11-4 TGL 200-3800.

8. Literaturverzeichnis

- /1/ Mikroprozessorsystem der II. Leistungsklasse, Hefte
CPU, PIO
VEB Mikroelektronik "Karl Marx" Erfurt
- /2/ Kieser, H. ; Meder, M. "Mikroprozessortechnik"
VEB Verlag Technik, Berlin 1986
- /3/ Kahl, B. " Mikroelektronik Information Applikation"
Heft 38 C570D + C571D Schnelle 8- und 10 Bit - Analog-
Digital-Wandler
Kammer der Technik, Vorstand des Bezirksverbandes Frank-
furt (Oder), Ebertusstraße 2, Frankfurt (O.) 1986
- /4/ Barthold, H. ; Bäurich, H. "Mikroprozessoren -
Mikroelektronische Schaltkreise und ihre Anwendung"
(Teil 2)
Reihe electronica Bd. 224/225
Militärverlag der DDR, Berlin 1980
- /5/ Kühnel, C. "AD - und DA -Umsetzer für Amateure"
Reihe electronica Bd. 232
Militärverlag der DDR, Berlin 1986
- /6/ Sieder, R. ; Kraft, D. ; Schenk, B. "Analogeingabemodul
M010 ADU1 für KC85/2 und KC85/3"
In: radio fernsehen elektronik 4/87, S. 253 und 254
- /7/ SEG Elektrotechnik/Elektronik/Automatisierungstechnik
Bedienanleitung und methodische Hinweise für den Lehrer
VEB Polytechnik Karl-Marx-Stadt

Wir danken dem Kollektiv der Zentralklinik für Herz- und
Lungenkrankheiten Bad Berka, Abt. Biomedizintechnik und
Informationsverarbeitung, für seine Mitarbeit bei der Entwicklung
des Moduls.

Gesamtherstellung

veb mikroelektronik "wilhelm pieck" mühlhausen

Ohne Genehmigung des Herausgebers ist es nicht gestattet, das
Heft oder Teile daraus nachzudrucken oder auf fotomechanischem
Wege zu vervielfältigen.